
#include <machine/asm.h>

LEAF(ffs, 1)
	addl	a0, 0, t0
	beq	t0, Lallzero

	/*
	 * Initialize return value (v0), and set up t1 so that it
	 * contains the mask with only the lowest bit set.
	 */
	subl	zero, t0, t1
	ldil	v0, 1
	and	t0, t1, t1

	and	t1, 0xff, t2
	bne	t2, Ldo8

	/*
	 * If lower 16 bits empty, add 16 to result and use upper 16.
	 */
	zapnot	t1, 0x03, t3
	bne	t3, Ldo16
	sra	t1, 16, t1
	addl	v0, 16, v0

Ldo16:
	/*
	 * If lower 8 bits empty, add 8 to result and use upper 8.
	 */
	and	t1, 0xff, t4
	bne	t4, Ldo8
	sra	t1, 8, t1
	addl	v0, 8, v0

Ldo8:
	and	t1, 0x0f, t5		/* lower 4 of 8 empty? */
	and	t1, 0x33, t6		/* lower 2 of each 4 empty? */
	and	t1, 0x55, t7		/* lower 1 of each 2 empty? */

	/* If lower 4 bits empty, add 4 to result. */
	bne	t5, Ldo4
	addl	v0, 4, v0

Ldo4:	/* If lower 2 bits of each 4 empty, add 2 to result. */
	bne	t6, Ldo2
	addl	v0, 2, v0

Ldo2:	/* If lower bit of each 2 empty, add 1 to result. */
	bne	t7, Ldone
	addl	v0, 1, v0

Ldone:
	RET

Lallzero:
	bis	zero, zero, v0
	RET
END(ffs)
